NAG Library Routine Document

F07HSF (ZPBTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07HSF (ZPBTRS) solves a complex Hermitian positive definite band system of linear equations with multiple right-hand sides,

$$AX = B,$$

where A has been factorized by F07HRF (ZPBTRF).

2 Specification

```
SUBROUTINE F07HSF (UPLO, N, KD, NRHS, AB, LDAB, B, LDB, INFO)
INTEGER                N, KD, NRHS, LDAB, LDB, INFO
COMPLEX (KIND=nag_wp) AB(LDAB,*), B(LDB,*)
CHARACTER(1)           UPLO
```

The routine may be called by its LAPACK name ***zpbtrs***.

3 Description

F07HSF (ZPBTRS) is used to solve a complex Hermitian positive definite band system of linear equations $AX = B$, the routine must be preceded by a call to F07HRF (ZPBTRF) which computes the Cholesky factorization of A. The solution X is computed by forward and backward substitution.

If UPLO = 'U', $A = U^{\mathrm{H}}U$, where U is upper triangular; the solution X is computed by solving $U^{\mathrm{H}}Y = B$ and then $UX = Y$.

If UPLO = 'L', $A = LL^{\mathrm{H}}$, where L is lower triangular; the solution X is computed by solving $LY = B$ and then $L^{\mathrm{H}}X = Y$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
$A = U^{\mathrm{H}}U$, where U is upper triangular.

UPLO = 'L'
$A = LL^{\mathrm{H}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: KD – INTEGER *Input*

On entry: k_d, the number of superdiagonals or subdiagonals of the matrix A.

Constraint: $\mathrm{KD} \geq 0$.

4: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: $\mathrm{NRHS} \geq 0$.

5: $\mathrm{AB}(\mathrm{LDAB},*)$ – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array AB must be at least $\max(1,\mathrm{N})$.

On entry: the Cholesky factor of A, as returned by F07HRF (ZPBTRF).

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HSF (ZPBTRS) is called.

Constraint: $\mathrm{LDAB} \geq \mathrm{KD}+1$.

7: $\mathrm{B}(\mathrm{LDB},*)$ – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1,\mathrm{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07HSF (ZPBTRS) is called.

Constraint: $\mathrm{LDB} \geq \max(1,\mathrm{N})$.

9: INFO – INTEGER *Output*

On exit: $\mathrm{INFO}=0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\mathrm{INFO}<0$

If $\mathrm{INFO}=-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A+E)x=b$, where

if UPLO = 'U', $|E| \leq c(k+1)\epsilon|U^{\mathrm{H}}||U|$;

if UPLO = 'L', $|E| \leq c(k+1)\epsilon|L||L^{\mathrm{H}}|$,

$c(k+1)$ is a modest linear function of $k+1$, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(k+1)\,\mathrm{cond}(A,x)\epsilon$$

where $\text{cond}(A,x) = \left\| |A^{-1}||A||x| \right\|_\infty / \|x\|_\infty \leq \text{cond}(A) = \left\| |A^{-1}||A| \right\|_\infty \leq \kappa_\infty(A)$. Note that $\text{cond}(A,x)$ can be much smaller than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07HVF (ZPBRFS), and an estimate for $\kappa_\infty(A)$ $(= \kappa_1(A))$ can be obtained by calling F07HUF (ZPBCON).

8 Parallelism and Performance

F07HSF (ZPBTRS) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07HSF (ZPBTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $16nkr$, assuming $n \gg k$.

This routine may be followed by a call to F07HVF (ZPBRFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07HEF (DPBTRS).

10 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} 9.39 + 0.00i & 1.08 - 1.73i & 0.00 + 0.00i & 0.00 + 0.00i \\ 1.08 + 1.73i & 1.69 + 0.00i & -0.04 + 0.29i & 0.00 + 0.00i \\ 0.00 + 0.00i & -0.04 - 0.29i & 2.65 + 0.00i & -0.33 + 2.24i \\ 0.00 + 0.00i & 0.00 + 0.00i & -0.33 - 2.24i & 2.17 + 0.00i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -12.42 + 68.42i & 54.30 - 56.56i \\ -9.93 + 0.88i & 18.32 + 4.76i \\ -27.30 - 0.01i & -4.40 + 9.97i \\ 5.31 + 23.63i & 9.43 + 1.41i \end{pmatrix}.$$

Here A is Hermitian positive definite, and is treated as a band matrix, which must first be factorized by F07HRF (ZPBTRF).

10.1 Program Text

```
    Program f07hsfe

!     F07HSF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zpbtrf, zpbtrs
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, j, kd, ldab, ldb, n, &
                                          nrhs
      Character (1)                    :: uplo
```

```
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,:), b(:,:)
      Character (1)                      :: clabs(1), rlabs(1)
!     .. Intrinsic Procedures ..
      Intrinsic                          :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07HSF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd, nrhs
      ldab = kd + 1
      ldb = n
      Allocate (ab(ldab,n),b(ldb,nrhs))

!     Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Factorize A
!     The NAG name equivalent of zpbtrf is f07hrf
      Call zpbtrf(uplo,n,kd,ab,ldab,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute solution
!       The NAG name equivalent of zpbtrs is f07hsf
        Call zpbtrs(uplo,n,kd,nrhs,ab,ldab,b,ldb,info)

!       Print solution

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4',            &
          'Solution(s)','Integer',rlabs,'Integer',clabs,80,0,ifail)

      Else
        Write (nout,*) 'A is not positive definite'
      End If

    End Program f07hsfe
```

10.2 Program Data

```
F07HSF Example Program Data
  4  1  2                                              :Values of N, KD and NRHS
  'L'                                                      :Value of UPLO
 ( 9.39, 0.00)
 ( 1.08, 1.73) ( 1.69, 0.00)
               (-0.04,-0.29) ( 2.65, 0.00)
                             (-0.33,-2.24) ( 2.17, 0.00)  :End of matrix A
 (-12.42,68.42) (54.30,-56.56)
 ( -9.93, 0.88) (18.32,  4.76)
 (-27.30,-0.01) (-4.40,  9.97)
 (  5.31,23.63) ( 9.43,  1.41)                             :End of matrix B
```

10.3 Program Results

```
F07HSF Example Program Results

Solution(s)
                    1                    2
1  (-1.0000, 8.0000) ( 5.0000,-6.0000)
2  ( 2.0000,-3.0000) ( 2.0000, 3.0000)
3  (-4.0000,-5.0000) (-8.0000, 4.0000)
4  ( 7.0000, 6.0000) (-1.0000,-7.0000)
```